



APPLE - BUG



Bi-Centennial Issue

The Newsletter of APPLE-Q Inc. the Brisbane Users' Group

Post Office Box 721 - South Brisbane - Qld 4101.

JANUARY 1988

[in this issue]

Editorial
Trading Table
Feature Article (continued...)

NEXT MEETING: - 17th January 1988

<<< CREDITS >>>

Graham Black -- stuck with it again
the Committee -- for their patience.
the Members -- for their tolerance
Microsoft Word -- Desktop Publishing.
EPSON LQ-1500 -- for the typesetting
El Presidente -- printing & distribution

[Executive Committee]

PRESIDENT	BERNIE BENSON	07-345-1545
VICE-PRESIDENT	ERIC CONOLLY	07-261-1860
SECRETARY	BOB GODBEHERE	07-808-3892
TREASURER	TONY TRUSCOTT	07-266-4795
REGISTRAR	GRAHAM BLACK	07-883-1525
EDITOR	{VACANT}	
ASSISTANT EDITOR	{VACANT}	
HARDCOPY LIBRARIAN	BRETT DUTTON	07-265-5112
SOFTWARE LIBRARIAN	BOB GODBEHERE	07-808-3892
(IN RESERVE).....	VINCE CROSDALE.	(having fun on the gold coast)

[Bulletin Board System (BBS)]

APPLE-Q Inc. BBS

NODE #1 : online 24 hours
Telephone : (07) 284-6146
baud Rates : 300,1200/75,1200,2400 (CCITT & BELL)
Data Specs : 8 Data bits 1 stop bit No Parity Full Duplex
Sysops : Graham Black, Brett Dutton & Vince Crosdale
Calls to system : 8140
Registered users : 358 (as at 21st December 1987)

NODE #2 : online Sunday to Friday 9:30pm - 5:30pm
Friday 9:30pm - noon Saturday
Saturday 9:30pm - 8:00am Sunday
Telephone : (07) 800-4660
Baud Rates : 300,1200/75,1200
Data Specs : 8 Data bits 1 stop bit No Parity Full Duplex
Sysop : Kelvin Saggars

SYSOP stands for : SYStem OPERator
BBS stands for : Bulletin Board System

[What's When]

At the Hooper Education Centre - Kuran Street - Wavell Heights - on the third (3rd) Sunday of each month, except December.

Sunday 17th January 1988 Open Day Hours: 9:00am - 4:30pm	Monday 18th January 1988 Committee Meeting Starts: 7:00pm
Sunday 21st February 1988 Open Day Hours: 9:00am - 4:30pm	Monday 22nd February 1988 Committee Meeting Starts: 7:00pm

[Membership Fees]

Joining Fee: \$10 Adults/Family: \$20
Students/Pensioners: \$12 Corporate Membership: \$50
Associate Membership: \$ 5 plus \$5 Joining Fee.(Pro-rata)
(students under 21 years - Pensioners on production of Pensioner Card
- at the discretion of the Executive Committee)

[Editorial]

<< COMPLIMENTS OF THE SEASON TO YOU ALL >>

The November Open Day went without a hitch, and the 'UNOFFICIAL DECEMBER OPEN DAY' which was held at Brett Dutton's place, was enjoyed by the few who attended. Thanks for the use of the pool, Brett. It is unfortunate that not many members attended, as the hire of tables was not cheap. Next year there will not be a December meeting, as not many members seem to be too interested.

Well the temperatures have been climbing steadily, and as most of you know, this can play havock with computers. The club's BBS is no exception to this phenomenon. On the 19th November the system went down several times, with the power supply overheating. This caused many problems for the Sysops. The file transfer went walk-about when the system crashed, and for several days the file transfer was unavailable to the users of the system. The computer went back for repairs, and is now back on-line, with a new power supply and motherboard.

We are in need of articles for APPLE-BUG. To date, everything has been left up to the same few people to fill these pages with interesting things for you to read. The trouble is that the same few people are running out of things to say each month. We would welcome anything and everything from the members to help fill these pages, even if it's just a thankyou for producing these pages. If you have a problem with something computer-related, then write it down, and we will include it in the next newsletter. The deadline for all submissions is the OPEN DAY, so bring your article, query, solution to a problem, or whatever, to the meeting and hand it to a committee member for inclusion in the next APPLE-BUG. We really won't laugh at any problem you might have, we simply want to help the members, after-all, that is why we all join clubs, to learn about our hobbies. I know that most people are only too happy to help, even if it's just to show how much they have learnt from asking questions themselves. If you are having a problem with something, then there will be others who have had the same or similar problems, and they probably now know the answer, so don't be too shy about asking for help. We all need it at some stage of the game.

Our Secretary (Vince Crosdale) has deserted us. He has at long last found a job, and he is now living somewhere on the Gold Coast. Vince has been with us for quite a few years now and has put in many a long hour working on club business. We are going to miss you on the committee, Vince. Anyway, he tells me he will still be coming to the meetings, when his new job permits, so we haven't lost him altogether. Bob Godbehere has taken on the job of Secretary, and now we need someone to take over Bob's jealously guarded position of Software Librarian. While I am at it, we are still looking for an Editor, and Assistant Editor. Would someone like to volunteer? It's getting to the stage, where if we don't find a volunteer soon, we won't have anyone to produce a newsletter.

Well, enough of my ranting and raving, I'll now 'Pass the Buck' and let the others type something, because I've run out of things to say.

<< IMPORTANT !! >>

Members are once again reminded that they must display their Membership Cards on their machines at the Open Days. This is to help us identify the NON-MEMBERS who are bludging on the club facilities. Many people seem to have forgotten the inconvenience caused by all those who would not pay their fees, taking up all the table space, so that members had nowhere to put their equipment. If you do not display your Membership Card, then don't complain when you are told to 'pay up or get out'.

[Brett's Speil]

by Brett Dutton

Since the last official meeting many things have happened. For a start, Vince has now got himself a job down the coast. This is good news for Vince but leaves the club without a secretary and Sysop. I have taken over the job of Sysop, and have started modifying selected sections of the software. This only leaves the position of secretary unfilled, but the good news is that Bob is going to take over that position, and you guessed it, nobody has the position of software librarian. I have been asked to do this job as well as Hard-Copy librarian, which is fine by me, except that I have since moved out of home, and the Apple I came to love had to stay (it was my fathers in the first place). Things went reasonably well at the last meeting in so far as the Hard-Copy Library is concerned but there are still many books out by members that MUST be returned. This is the last mention of this before names will be published. At the next meeting there will be a selected number of books and magazines on display and borrowing will take place with the rules, mentioned in the last newsletter, in mind. The Christmas meeting at my place was a complete flop seven people showed up. Tables were hired and in my mind money was completely wasted. Next year there will be no Christmas meeting. Despite the poor turnout a good time was had by all those who attended.

Have a very Merry Christmas and a Happy New Year.

[Trading Table]

by Graham Black

The Trading Table will be operating once again in the new year. We will have disks and disk boxes on sale as usual.

We now have CREDIT CARD facilities available. You may now use your Mastercard, Bankcard or Visa to purchase items, or to RENEW YOUR MEMBERSHIP on the Open Days.

JANUARY SPECIAL

*
* We have been able to get our hands on some *
* disks at a cheaper price than usual, and *
* these will be on sale at the January meeting *
* at \$11.00 per box of 10. *
*
* FIRST IN FIRST SERVED *
*

[Feature Article]

by Cecil Fretwell

AN APPLEWORKS UTILITY (continued

Permission to publish this article in APPLE-BUG has been granted by Cecil Fretwell (the author). This article is Copywrite (C).

```

100 REM FAW.UTILITY
110 REM
120 REM *****
130 REM * AN APPLEWORKS UTILITY *
140 REM * BY *
150 REM * CECIL FRETWELL *
160 REM * 6-MAY-86 *
170 REM *****
180 REM
190 REM *****
200 REM * INSTALL AMPERSAND FUNCTION *
210 REM * TO DELETE AN ARRAY *
220 REM *****
230 REM
240 FOR I = 752 TO 827: READ J: POKE I,J: NEXT
250 CALL 752
260 DATA 162,2,189,253,2,157,245,3,202,16,247,24,96,76,0,3,32
270 DATA 183,0,201,214,240,3,76,201,222,32,177,0,32,217,247,160,2
280 DATA 24,165,155,133,66,113,155,133,60,165,156,133,67,200,113
290 DATA 155,133,61,56,165,109,133,62,136,241,155,133,109,165,110
300 DATA 133,63,200,241,155,133,110,160,00,76,44,254
310 REM
320 REM *****
330 REM * MAIN MENU STARTS HERE *
340 REM *****
350 REM
360 TEXT : HOME : POKE 216,0: CLEAR : PRINT CHR$(4);"PR#3"
370 MS$ = "AN APPLEWORKS UTILITY": GOSUB 3860: PRINT
380 MS$ = "BY": GOSUB 3860: PRINT
390 MS$ = "CECIL FRETWELL": GOSUB 3860: PRINT
400 MS$ = "6-MAY-86": GOSUB 3860: PRINT : PRINT
410 MS$ = "1. DISPLAY CATALOG OF A DISK": GOSUB 3860: PRINT
420 MS$ = "2. PRINT CATALOG OF A DISK ": GOSUB 3860: PRINT
430 MS$ = "3. LOCK A FILE ": GOSUB 3860: PRINT
440 MS$ = "4. UNLOCK A FILE ": GOSUB 3860: PRINT
450 MS$ = "5. START UP APPLEWORKS ": GOSUB 3860: PRINT
460 MS$ = "6. EXIT TO APPLESOFT ": GOSUB 3860: PRINT
470 PRINT : HTAB 26: VTAB 13
480 MS$ = "PLEASE SELECT AN OPTION: ": PRINT MS$;
490 GET A$: PRINT A$
500 IF A$ = CHR$(3) THEN 360
510 IF A$ = "1" THEN PF$ = "N": GOTO 650
520 IF A$ = "2" THEN PF$ = "Y": GOTO 650
530 IF A$ = "3" THEN 4180
540 IF A$ = "4" THEN 4430
550 IF A$ = "5" THEN 4680
560 IF A$ = "6" THEN TEXT : HOME : END
570 PRINT CHR$(7); CHR$(7); CHR$(7);
580 GOTO 470
600 REM *****
610 REM * 1. DISPLAY CATALOG OF A DISK *
620 REM * 2. PRINT CATALOG OF A DISK *

```

```

630 REM *****
640 REM
650 ONERR GOTO 3920
660 HTAB 15: VTAB 15
670 PRINT "ENTER DESIRED VOLUME NAME OR DIRECTORY NAME BELOW"
680 HTAB 15: VTAB 16
690 INPUT " ";PX$
700 IF PX$ = "" THEN 360
710 IF LEFT$(PX$,1) = "/" THEN 730
720 PRINT CHR$(4);"PREFIX": INPUT T$:PX$ = T$ + PX$
730 D = 0
740 IF RIGHT$(PX$,1) < > "/" THEN PX$ = PX$ + "/"
750 PRINT CHR$(4);"VERIFY ";PX$
760 IF PF$ = "N" THEN HOME : GOTO 820
770 HTAB 15: VTAB 18:PA% = 0
780 PRINT "READY YOUR PRINTER THEN PRESS ANY KEY";
790 GET A$: PRINT A$
800 IF A$ = CHR$(3) THEN 360
810 PRINT CHR$(4);"PR#1"
820 PRINT CHR$(4);"OPEN ";PX$;" ,TDIR"
830 PRINT CHR$(4);"READ ";PX$
840 D = PEEK (116) * 256 + 1024
850 D = PEEK (D + 37) + PEEK (D + 38) * 256
860 DIM R$(D)
870 INPUT RN$: INPUT T$: INPUT T$
880 R = 1
890 INPUT T$: IF T$ = "" THEN 890
900 IF LEFT$(T$,12) = "BLOCKS FREE:" THEN 950
910 GOSUB 1540
920 R$(R) = N$ + TY$ + MID$(T$,30,10) + LI$ + DL$
930 R = R + 1
940 GOTO 890
950 R = R - 1
960 IF R = 0 THEN PRINT PX$;" CONTAINS NO FILES": END
970 BF$ = T$
980 PRINT CHR$(4);"CLOSE"
990 Z = 1
1000 Z = 3 * Z + 1: IF Z < R THEN 1000
1010 Z = (Z - 1) / 3
1020 IF Z < 1 THEN 1840
1030 FOR I = Z TO R
1040 J = I - Z
1050 K = J + Z
1060 IF MID$(R$(J),2,15) < MID$(R$(K),2,15) THEN 1090
1070 T$ = R$(J):R$(J) = R$(K):R$(K) = T$
1080 J = J - Z: IF J > = 1 THEN 1050
1090 NEXT I
1100 GOTO 1010
1110 REM
1120 REM *****
1130 REM * SUBROUTINE TO PAUSE SCREEN OR FORM FEED PRINTER *
1140 REM *****
1150 REM
1160 LI% = LI% + 1
1170 IF PF$ = "Y" THEN 1250
1180 IF LI% < > 20 THEN RETURN
1190 LI% = 0
1200 PRINT : HTAB 26: PRINT "PRESS ANY KEY";
1210 Z% = PEEK (- 16384): IF Z% < 128 THEN 1210
1220 POKE - 16368,0: PRINT
1230 CALL 64538: CALL 64668: CALL 64538

```

```

1240 RETURN
1250 IF LI% < > 60 THEN RETURN
1260 PRINT CHR$(12);"CATALOG OF ";PX$;
1270 IF PEEK(49040) = 0 THEN 1450
1280 DA = PEEK(49041) * 256 + PEEK(49040)
1290 YR = INT(DA / 512):DA = DA - YR * 512
1300 MO = INT(DA / 32):DA = DA - MO * 32
1310 PRINT " AS OF ";DA;"-";
1320 IF MO = 1 THEN PRINT "JAN";
1330 IF MO = 2 THEN PRINT "FEB";
1340 IF MO = 3 THEN PRINT "MAR";
1350 IF MO = 4 THEN PRINT "APR";
1360 IF MO = 5 THEN PRINT "MAY";
1370 IF MO = 6 THEN PRINT "JUN";
1380 IF MO = 7 THEN PRINT "JUL";
1390 IF MO = 8 THEN PRINT "AUG";
1400 IF MO = 9 THEN PRINT "SEP";
1410 IF MO = 10 THEN PRINT "OCT";
1420 IF MO = 11 THEN PRINT "NOV";
1430 IF MO = 12 THEN PRINT "DEC";
1440 PRINT "-";YR;
1450 IF PF$ = "N" THEN PRINT : GOTO 1470
1460 PA% = PA% + 1: PRINT " PAGE ";PA%
1470 PRINT :LI% = 2
1480 RETURN
1490 REM
1500 REM *****
1510 REM * SUBROUTINE TO BUILD CATALOG ENTRY *
1520 REM *****
1530 REM
1540 N$ = MID$(T$,1,16):TY$ = MID$(T$,18,3)
1550 IF TY$ = "DIR" THEN TY$ = " DIRECTORY "
1560 IF TY$ = "SYS" THEN TY$ = " SYSTEM PROGRAM"
1570 IF TY$ = "BAS" THEN TY$ = " BASIC PROGRAM "
1580 IF TY$ = "BIN" THEN TY$ = " BINARY PROGRAM"
1590 IF TY$ = "TXT" THEN TY$ = " TEXT "
1600 IF TY$ = "AWP" THEN TY$ = " WORD PROCESSOR"
1610 IF TY$ = "ASP" THEN TY$ = " SPREADSHEET "
1620 IF TY$ = "ADB" THEN TY$ = " DATA BASE "
1630 IF LEN(TY$) = 3 THEN TY$ = " " + TY$ + " "
1640 LI$ = MID$(T$,25,4)
1650 IF LEFT$(LI$,3) < > " " THEN 1680
1660 LI = VAL(RIGHT$(LI$,1)) / 2
1670 GOTO 1750
1680 IF LEFT$(LI$,3) < > " " THEN 1710
1690 LI = VAL(RIGHT$(LI$,2)) / 2
1700 GOTO 1750
1710 IF LEFT$(LI$,3) < > " " THEN 1740
1720 LI = VAL(RIGHT$(LI$,3)) / 2
1730 GOTO 1750
1740 LI = VAL(LI$) / 2
1750 IF TY$ = " DIRECTORY " THEN LI$ = " ": GOTO 1770
1760 LI$ = RIGHT$(" " + STR$(LI) + "K",8)
1770 DL$ = STR$(LI)
1780 RETURN
1790 REM
1800 REM *****
1810 REM * ROOT ARRAY IS NOW BUILT AND SORTED *
1820 REM *****
1830 REM
1840 PRINT CHR$(4);"FRE"

```

```

1850 GOSUB 1260
1860 FOR RI = 1 TO R
1870 FOR I = 2 TO 16
1880 IF MID$(R$(RI),I,1) = " " THEN PRINT ".": GOTO 1900
1890 PRINT MID$(R$(RI),I,1);
1900 NEXT I
1910 PRINT "....."; MID$(R$(RI),17,33);
1920 IF LEFT$(R$(RI),1) = "*" THEN PRINT " LOCKED";
1930 PRINT : GOSUB 1160
1940 IF MID$(R$(RI),18,9) = "DIRECTORY" THEN 2230
1950 NEXT RI
1960 & FRE R$
1970 PRINT : PRINT "THE ENTIRE DISK HAS ";
1980 FR$ = MID$(BF$,14,4)
1990 IF LEFT$(FR$,3) = " " THEN FR$ = RIGHT$(FR$,1): GOTO 2020
2000 IF LEFT$(FR$,2) = " " THEN FR$ = RIGHT$(FR$,2): GOTO 2020
2010 IF LEFT$(FR$,1) = " " THEN FR$ = RIGHT$(FR$,3)
2020 FR = VAL (FR$) / 2:FR$ = STR$ (FR)
2030 PRINT FR;"K FREE"; SPC( 8 - LEN (FR$));
2040 US$ = MID$(BF$,36,4)
2050 IF LEFT$(US$,3) = " " THEN US$ = RIGHT$(US$,1): GOTO 2080
2060 IF LEFT$(US$,2) = " " THEN US$ = RIGHT$(US$,2): GOTO 2080
2070 IF LEFT$(US$,1) = " " THEN US$ = RIGHT$(US$,3)
2080 US = VAL (US$) / 2:US$ = STR$ (US)
2090 PRINT US;"K USED"; SPC( 8 - LEN (US$));FR + US;"K TOTAL"
2100 REM
2110 REM *****
2120 REM * END OF ROOT PROCESSING *
2130 REM *****
2140 REM
2150 IF PF$ = "N" THEN GOSUB 1200: GOTO 360
2160 PRINT CHR$(4);"PR#0"
2170 GOTO 360
2180 REM
2190 REM *****
2200 REM * PROCESS FIRST LEVEL SUB DIRECTORY *
2210 REM *****
2220 REM
2230 D1 = VAL ( MID$( R$(RI),50)) * 2
2240 DIM D1$(13 * D1)
2250 ND$ = PX$
2260 FOR I = 2 TO 16
2270 IF MID$(R$(RI),I,1) < > " " THEN ND$ = ND$ + MID$(R$(RI),I,1)
2280 NEXT I
2290 PRINT CHR$(4);"OPEN ";ND$;" ,TDIR"
2300 PRINT CHR$(4);"READ ";ND$
2310 INPUT T$: INPUT T$: INPUT T$
2320 L1 = 1
2330 INPUT T$: IF T$ = "" THEN 2330
2340 IF LEFT$(T$,12) = "BLOCKS FREE:" THEN 2390
2350 GOSUB 1540
2360 D1$(L1) = N$ + TY$ + MID$(T$,30,10) + LI$ + DL$
2370 L1 = L1 + 1
2380 GOTO 2330
2390 PRINT CHR$(4);"CLOSE"
2400 L1 = L1 - 1
2410 IF L1 = 0 THEN 2660
2420 Z = 1
2430 Z = 3 * Z + 1: IF Z < L1 THEN 2430
2440 Z = (Z - 1) / 3
2450 IF Z < 1 THEN 2550

```

```

2460 FOR I = Z TO L1
2470 J = I - Z
2480 K = J + Z
2490 IF MID$ (D1$(J),2,15) < MID$ (D1$(K),2,15) THEN 2530
2500 T$ = D1$(J):D1$(J) = D1$(K):D1$(K) = T$
2510 J = J - Z
2520 IF J > = 1 THEN 2480
2530 NEXT I
2540 GOTO 2440
2550 FOR D1 = 1 TO L1
2560 PRINT " ";
2570 FOR I = 2 TO 16
2580 IF MID$ (D1$(D1),I,1) = " " THEN PRINT ".": GOTO 2600
2590 PRINT MID$ (D1$(D1),I,1);
2600 NEXT I
2610 PRINT "....."; MID$ (D1$(D1),17,33);
2620 IF LEFT$ (D1$(D1),1) = "*" THEN PRINT " LOCKED";
2630 PRINT : GOSUB 1160
2640 IF MID$ (D1$(D1),18,9) = "DIRECTORY" THEN 2780
2650 NEXT D1
2660 & FRE D1$
2670 PRINT CHR$ (4);"FRE"
2680 GOTO 1950
2690 REM
2700 REM *****
2710 REM * END OF FIRST LEVEL SUB DIRECTORY PROCESSING *
2720 REM *****
2730 REM
2740 REM *****
2750 REM * PROCESS SECOND LEVEL SUB DIRECTORY *
2760 REM *****
2770 REM
2780 D2 = VAL ( MID$ (D1$(D1),50)) * 2
2790 DIM D2$(13 * D2)
2800 L1$ = ND$ + "/"
2810 FOR I = 2 TO 16
2820 IF MID$ (D1$(D1),I,1) < > " " THEN L1$ = L1$ + MID$ (D1$(D1),I,1)
2830 NEXT I
2840 PRINT CHR$ (4);"OPEN ";L1$;" ,TDIR"
2850 PRINT CHR$ (4);"READ ";L1$
2860 INPUT T$: INPUT T$: INPUT T$
2870 L2 = 1
2880 INPUT T$: IF T$ = "" THEN 2880
2890 IF LEFT$ (T$,12) = "BLOCKS FREE:" THEN 2940
2900 GOSUB 1540
2910 D2$(L2) = N$ + TY$ + MID$ (T$,30,10) + L1$ + D1$
2920 L2 = L2 + 1
2930 GOTO 2880
2940 PRINT CHR$ (4);"CLOSE"
2950 L2 = L2 - 1
2960 IF L2 = 0 THEN 3200
2970 Z = 1
2980 Z = 3 * Z + 1: IF Z < L2 THEN 2980
2990 Z = (Z - 1) / 3
3000 IF Z < 1 THEN 3090
3010 FOR I = Z TO L2
3020 J = I - Z
3030 K = J + Z
3040 IF MID$ (D2$(J),2,15) < MID$ (D2$(K),2,15) THEN 3070
3050 T$ = D2$(J):D2$(J) = D2$(K):D2$(K) = T$
3060 J = J - Z: IF J > = 1 THEN 3030

```

```

3070 NEXT I
3080 GOTO 2990
3090 FOR D2 = 1 TO L2
3100 PRINT " ";
3110 FOR I = 2 TO 16
3120 IF MID$(D2$(D2),I,1) = " " THEN PRINT ".": GOTO 3140
3130 PRINT MID$(D2$(D2),I,1);
3140 NEXT I
3150 PRINT "....."; MID$(D2$(D2),17,33);
3160 IF LEFT$(D2$(D2),1) = "*" THEN PRINT " LOCKED";
3170 PRINT : GOSUB 1160
3180 IF MID$(D2$(D2),18,9) = "DIRECTORY" THEN 3320
3190 NEXT D2
3200 & FRE D2$
3210 PRINT CHR$(4);"FRE"
3220 GOTO 2650
3230 REM
3240 REM *****
3250 REM * END OF SECOND LEVEL SUB DIRECTORY PROCESSING *
3260 REM *****
3270 REM
3280 REM *****
3290 REM * PROCESS THIRD LEVEL SUB DIRECTORY *
3300 REM *****
3310 REM
3320 D3 = VAL ( MID$( D2$(D2),50) ) * 2
3330 DIM D3$(13 * D3)
3340 L2$ = L1$ + "/"
3350 FOR I = 2 TO 16
3360 IF MID$( D2$(D2),I,1) < > " " THEN L2$ = L2$ + MID$( D2$(D2),I,1)
3370 NEXT I
3380 PRINT CHR$(4);"OPEN ";L2$;" ,TDIR"
3390 PRINT CHR$(4);"READ ";L2$
3400 INPUT T$: INPUT T$: INPUT T$
3410 L3 = 1
3420 INPUT T$: IF T$ = "" THEN 3420
3430 IF LEFT$( T$,12) = "BLOCKS FREE:" THEN 3480
3440 GOSUB 1540
3450 D3$(L3) = N$ + TY$ + MID$( T$,30,10) + LI$ + DL$
3460 L3 = L3 + 1
3470 GOTO 3420
3480 PRINT CHR$(4);"CLOSE"
3490 L3 = L3 - 1
3500 IF L3 = 0 THEN 3730
3510 Z = 1
3520 Z = 3 * Z + 1: IF Z < L3 THEN 3520
3530 Z = (Z - 1) / 3
3540 IF Z < 1 THEN 3630
3550 FOR I = Z TO L3
3560 J = I - Z
3570 K = J + Z
3580 IF MID$( D3$(J),2,15) < MID$( D3$(K),2,15) THEN 3610
3590 T$ = D3$(J):D3$(J) = D3$(K):D3$(K) = T$
3600 J = J - Z: IF J > = 1 THEN 3570
3610 NEXT I
3620 GOTO 3530
3630 FOR D3 = 1 TO L3
3640 PRINT " ";
3650 FOR I = 2 TO 16
3660 IF MID$( D3$(D3),I,1) = " " THEN PRINT ".": GOTO 3680
3670 PRINT MID$( D3$(D3),I,1);

```

```

3680 NEXT I
3690 PRINT "....."; MID$(D3$(D3),17,33);
3700 IF LEFT$(D3$(D3),1) = "*" THEN PRINT " LOCKED";
3710 PRINT : GOSUB 1160
3720 NEXT D3
3730 & FRE D3$
3740 PRINT CHR$(4);"FRE"
3750 GOTO 3190
3760 REM
3770 REM *****
3780 REM * END OF THIRD LEVEL SUB DIRECTORY PROCESSING *
3790 REM *****
3800 REM
3810 REM *****
3820 REM * SUBROUTINE TO CENTER MESSAGE *
3830 REM * ON 80 COLUMN SCREEN *
3840 REM *****
3850 REM
3860 PRINT SPC( INT ((80 - LEN (MS$)) / 2));MS$;: RETURN
3870 REM
3880 REM *****
3890 REM * ONERR ROUTINE FOR CATALOG PROCESS *
3900 REM *****
3910 REM
3920 E$ = PEEK (222): POKE 216,0: CALL - 3288
3930 PRINT CHR$(4);"PR#0"
3940 PRINT CHR$(4);"CLOSE"
3950 IF E$ = 255 THEN 360
3960 PRINT : PRINT CHR$(7); CHR$(7); CHR$(7);
3970 HTAB 15: GOSUB 4050: IF D < > 0 THEN & FRE R$
3980 PRINT : HTAB 15: PRINT "PRESS ANY KEY";: GET A$: PRINT A$
3990 GOTO 650
4000 REM
4010 REM *****
4020 REM * SUBROUTINE TO DISPLAY ERROR MESSAGE *
4030 REM *****
4040 REM
4050 IF E$ = 4 THEN PRINT "DISK IS WRITE PROTECTED": RETURN
4060 IF E$ = 6 THEN PRINT "SPECIFIED VOLUME OR FILE NOT FOUND": RETURN
4070 IF E$ = 7 THEN PRINT "SPECIFIED VOLUME OR FILE NOT FOUND": RETURN
4080 IF E$ = 8 THEN PRINT "DISK I/O ERROR": RETURN
4090 IF E$ = 10 THEN PRINT "SPECIFIED FILE IS LOCKED": RETURN
4100 IF E$ = 16 THEN PRINT "ILLEGAL CHARACTER IN NAME": RETURN
4110 PRINT "ERROR ";E$;" LINE ";PEEK (218) + PEEK (219) * 256
4120 RETURN
4130 REM
4140 REM *****
4150 REM * 3. LOCK A FILE *
4160 REM *****
4170 REM
4180 ONERR GOTO 4330
4190 HTAB 15: VTAB 15: PRINT "ENTER DESIRED FILE NAME BELOW"
4200 HTAB 15: VTAB 16: INPUT "";PN$
4210 IF PN$ = "" THEN 360
4220 IF LEFT$(PN$,1) = "/" THEN 4240
4230 PRINT CHR$(4);"PREFIX": INPUT T$:PN$ = T$ + PN$
4240 HTAB 15: VTAB 18: CALL - 868
4250 PRINT CHR$(4);"VERIFY ";PN$
4260 PRINT CHR$(4);"LOCK ";PN$
4270 GOTO 360
4280 REM

```

```
4290 REM *****
4300 REM * ONERR ROUTINE FOR LOCK A FILE *
4310 REM *****
4320 REM
4330 E% = PEEK (222): POKE 216,0: CALL - 3288
4340 IF E% = 255 THEN 360
4350 HTAB 15: VTAB 18: PRINT CHR$ (7); CHR$ (7); CHR$ (7);
4360 GOSUB 4050
4370 GOTO 4180
4380 REM
4390 REM *****
4400 REM * 4. UNLOCK A FILE *
4410 REM *****
4420 REM
4430 ONERR GOTO 4580
4440 HTAB 15: VTAB 15: PRINT "ENTER DESIRED FILE NAME BELOW"
4450 HTAB 15: VTAB 16: INPUT "";PN$
4460 IF PN$ = "" THEN 360
4470 IF LEFT$ (PN$,1) = "/" THEN 4490
4480 PRINT CHR$ (4);"PREFIX": INPUT T$:PN$ = T$ + PN$
4490 HTAB 15: VTAB 18: CALL - 868
4500 PRINT CHR$ (4);"VERIFY ";PN$
4510 PRINT CHR$ (4);"UNLOCK ";PN$
4520 GOTO 360
4530 REM
4540 REM *****
4550 REM * ONERR ROUTINE FOR UNLOCK A FILE *
4560 REM *****
4570 REM
4580 E% = PEEK (222): POKE 216,0: CALL - 3288
4590 IF E% = 255 THEN 360
4600 HTAB 15: VTAB 18: PRINT CHR$ (7); CHR$ (7); CHR$ (7);
4610 GOSUB 4050
4620 GOTO 4430
4630 REM
4640 REM *****
4650 REM * 5. START UP APPLEWORKS *
4660 REM *****
4670 REM
4680 ONERR GOTO 4780
4690 HTAB 15: VTAB 15: PRINT "ENTER VOLUME CONTAINING APPLEWORKS"
4700 HTAB 15: VTAB 16: INPUT "";PN$
4710 PRINT CHR$ (4);"VERIFY ";PN$;"/APPLEWORKS/APLWORKS.SYSTEM"
4720 PRINT CHR$ (4);"-";PN$;"/APPLEWORKS/APLWORKS.SYSTEM"
4730 REM
4740 REM *****
4750 REM * ONERR ROUTINE FOR START UP APPLEWORKS *
4760 REM *****
4770 REM
4780 E% = PEEK (222): POKE 216,0: HTAB 15: VTAB 18: CALL - 3288
4790 IF E% = 255 THEN 360
4800 HTAB 15: VTAB 18
4810 IF E% < > 6 THEN 4850
4820 PRINT CHR$ (7); CHR$ (7); CHR$ (7);
4830 PRINT "DISK DOES NOT CONTAIN APPLEWORKS"
4840 GOTO 4860
4850 GOSUB 4050
4860 HTAB 15: VTAB 20: PRINT "PRESS ANY KEY";: GET A$: PRINT A$
4870 GOTO 360
```

President's Keyboard

Our Newsletter this month is somewhat different to the usual in that it was prepared using a photo copier as the Hooper Centre staff who usually print the magazine have not yet returned from the annual break. The presentation is not up to the same standard however it is hoped that you agree that it is better than none at all.

This issue is the clubs contribution to the bicentennial year celebrations and whilst we have not featured any 200 year old computer articles we have instead shown the Aussie colours.

To give you something to do with your spare time we would like your interpretation of the stylised Australian symbol being used by the bicentennial committee. The winning entry will be featured on a later edition of the newsletter. The size is to be restricted to the approximate size of the Apple on the cover of the magazine. For the month it is featured it will take the place of the Apple and Demon (I don't know what else you could call it).

Selection of the winning entry will be at the discretion of the Editor and if you feel you may not agree with his choice nominate for the position and you will only have yourself to blame if you don't win.

This month we complete the article by Cecil Fretwell and if you are not into Appleworks then maybe this issue may be of some benefit to you in studying some of the routines used and taking note of the presentation of the program itself.

Power Spike is missing from this issue and it is doubtful that he will return to print. So how about an article or two from some of you members out there.

Beagle Bros. Time Out - Superfonts

If you have this latest program from Beagle Bros. and are having trouble getting the printing right i.e. My copy failed to print the eighth printing wire on each pass.

Details of the fix are as follows:

Boot Superfont disk (not the original)

Select 3

Select 3

You are now in basic.

Type in the following:

POKE 768,52 < return >

BSAYE TO.SUPERFONTS,A768,L1,B46286 < return >

POKE 768,51 < return >

BSAYE TO.SUPERFONTS,A768,L1,B46342 < return >

The program should now print correctly if you had the same problem as mine.

This and the next page were printed using the Timeout Superfonts program.

Apple Bouquet

During the month we heard from a member who had received less than satisfactory service (out of pocket almost \$300) from three local Apple dealers regarding what appeared to be a printer card problem. As a last resort the member contacted the Brisbane office of Apple Australia and the member no longer had a problem - Apple located the fault - it was the cable - repaired same and restored the members faith in all things Apple.

We are not suggesting that you should contact Apple with all your problems however in this case I believe the member had a right to more professional service than was forthcoming from the dealers. Apple deserves a pat on the back.

B.B.S.

The club maintains an excellent Bulletin Board Service and in recent months with the assistance of Kelvin Sagers (he owns the equipment) we have endeavoured to establish a second board located on the south side to avoid the cost of S.T.D. calls for the members who live on the southern side of the river.

Initially we tried to tie the two systems together as regards to user base and this has proved to have been a mistake. We are now operating the board using a separate user base and you will have a different user number for each board.

Yours membership still applies to both boards.

If you have had problems with the south board in the past give it a go in 88 it should be a better experience.

New Game

B.B.S. users - Northside Board only:

**** Commander in Chief **** - Log on and give it a go!

Enhancing a IIe

An Enhancement Kit is available from Apple dealers to upgrade the early model IIe to include many of the features found in the IIc.

Dealer installation is required and your old chips are retained by the dealer.

To me this is giving away something you have already purchased plus if you at a later time wanted to revert to the original this will not be possible. If you wish to upgrade and retain your old chips it is suggested that you purchase an Enhancement kit from a U.K. dealer and replace the chips yourself.

Do not buy from U.S.A. as the Video Rom supplied with the U.S.A. kit is not suitable for our machines.

Dealer Installation Instructions come with the kit and with a little care and a few minutes work the job is complete.